Problem Set A

Problem 1

- **(a)**

```
N[413 / (768 + 295)]
```

```
0.388523
```

- **(b)**

```
2^123
```

```
10 633 823 966 279 326 983 230 456 482 242 756 608
```

```
N[%]
```

1.06338×10^{37}

- **(c)**

```
N[Pi^2, 35]
```

```
9.8696044010893586188344909998761511
```

```
N[E, 35]
```

```
2.7182818284590452353602874713526625
```

- **(d)**

```
N[{61 / 88, 13 863 / 20 000, 253 / 365, Log[2]}]
```

```
{0.693182, 0.69315, 0.693151, 0.693147}
```

The best approximation to log(2) is given by the second fraction.

Problem 2

- **(a)**

```
N[10 * Sin[1 / 10], 15]
```

```
0.998334166468282
```

- **(b)**

```
N[100 * Sin[1 / 100], 15]
```

```
0.999983333416666
```

- **(c)**

```
1000 * N[Sin[1 / 1000], 15]
```

```
0.999999833333342
```

This illustrates the fact that lim(x sin(1/x) =1 as x->infty.

Problem 3

```
? Plot
```

Plot[f, {x, x_{min}, x_{max}}] generates a plot of f as a function of x from x_{min} to x_{max}.
Plot[{f_1, f_2, …}, {x, x_{min}, x_{max}}] plots several functions f_i. ≫

- **(a)**

```
Plot[ x^3 - x, {x, -1.5, 1.5}]
```



- **(b)**

```
Plot[Tan[x], {x, -2 * Pi, 2 * Pi}]
```



(c)

```
? ContourPlot
```

ContourPlot[f, {x, x_{min}, x_{max}}, {y, y_{min}, y_{max}}]
generates a contour plot of f as a function of x and y.
ContourPlot[$f == g$, {x, x_{min}, x_{max}}, {y, y_{min}, y_{max}}] plots contour lines for which $f = g$.
ContourPlot[{$f_1 == g_1$, $f_2 == g_2$, …}, {x, x_{min}, x_{max}}, {y, y_{min}, y_{max}}]
plots several contour lines. ≫

```
ContourPlot[y^2 == x^3 - x, {x, -2.5, 2.5}, {y, -4, 4}]
```



Problem 4

- **(a)**

```
? Factor
```

Factor[*poly*] factors a polynomial over the integers.
Factor[*poly*, Modulus -> *p*] factors a polynomial modulo a prime *p*.
Factor[*poly*, Extension -> {a_1, a_2, ...}] factors a polynomial allowing coefficients that are rational combinations of the algebraic numbers a_i. ≫

```
Factor[x^3 + 5 * x^2 - 17 * x - 21]
```

```
(-3 + x) (1 + x) (7 + x)
```

- **(b)**

```
? FactorInteger
```

FactorInteger[*n*] gives a list of the prime factors of the integer *n*, together with their exponents.
FactorInteger[*n*, *k*] does partial factorization, pulling out at most *k* distinct factors. ≫

```
FactorInteger[123 456 789]
```

```
{{3, 2}, {3607, 1}, {3803, 1}}
```

Problem 5

In this problem, we find the intersections of the graphs of the functions x^8 and 4^x. First we graph the functions from 0 to 3.

```
Plot[{x^8, 4^x}, {x, 0, 3}]
```



It looks like there might be a point of intersection between 0 and 2, so let's zero in on that region.

```
Plot[{x^8, 4^x}, {x, 0, 2}]
```



```
Plot[{x^8, 4^x}, {x, 1.2, 1.3}]
```



It defintely looks like there's a point of intersection at about 1.24. Are there any other points of intersection? In the first plot it's hard to see what's happening at 0 (because the vertical scale is so large, from 0 to 500). Let's take a closer look at a neighborhood of 0.

```
Plot[{x^8, 4^x}, {x, -1, 1}]
```



Now we see that there appears to be a point of intersection near the point -0.85. Are there any other points of intersection? As x goes to -infty, the graph of x^8 goes off to +infty, but the graph of 4^x decays to zero, so there are no points of intersection to the left of the one we just iidentified at approx -0.85. Are ther any to the right of the one identified above at approx 1.24? In the graph above, the function whose graph passes through the point (0,1) is 4^x**.** In the very first graph of these two functions it appears that as x gets larger, x^8 remains larger than 4^x**.** We can use *Mathematica* to check whether this is true.

```
Limit[x^8 / 4^x, x -> Infinity]
```

0

So certainly 4^x eventually is bigger than x^8.

```
Plot[{x^8, 4^x}, {x, 2, 20}]
```



There it is. Let's zero in on it.

```
Plot[{x^8, 4^x}, {x, 15, 17}]
```



It looks like the intersection is pretty close to $x = 16$. Now let's use the command "FindRoot" to numerically find these points of intersection. Here's the syntax for this command:

```
? FindRoot
```

FindRoot[f, { x, x_0 }] searches for a numerical root of f, starting from the point $x = x_0$.
FindRoot[$lhs == rhs$, { x, x_0 }] searches for a numerical solution to the equation $lhs == rhs$.
FindRoot[{ f_1, f_2, …}, {{ x, x_0 }, { y, y_0 }, …}]
searches for a simultaneous numerical root of all the f_i.
FindRoot[{ eqn_1, eqn_2, …}, {{ x, x_0 }, { y, y_0 }, …}] searches for a
numerical solution to the simultaneous equations eqn_i. ≫

This command looks for the "closest" root to the starting point x_0. To find the three different points of intersection we will have to use three different starting points.

```
FindRoot[x^8 == 4^x, {x, -1}]
```

```
{x → -0.861345}
```

```
FindRoot[x^8 == 4^x, {x, 1}]
```

```
{x → 1.23963}
```

```
FindRoot[x^8 == 4^x, {x, 16}]
```

```
{x → 16.}
```

Those are the three values (at least to five decimal places) where the two functions x^8 and 4^x have the same value. Just for the record, here is the unhelpful output using **Solve**.

```
Solve[x^8 == 4^x, x]
```

Solve::ifun : Inverse functions are being used by Solve, so some solutions may not be found; use Reduce for complete solution information. ≫

$$\left\{\left\{x \to -\frac{8\,\text{ProductLog}\left[-\frac{\text{Log}[4]}{8}\right]}{\text{Log}[4]}\right\}, \left\{x \to -\frac{8\,\text{ProductLog}\left[-\frac{1}{8}\,i\,\text{Log}[4]\right]}{\text{Log}[4]}\right\},\right.$$

$$\left\{x \to -\frac{8\,\text{ProductLog}\left[\frac{1}{8}\,i\,\text{Log}[4]\right]}{\text{Log}[4]}\right\}, \left\{x \to -\frac{8\,\text{ProductLog}\left[\frac{\text{Log}[4]}{8}\right]}{\text{Log}[4]}\right\},$$

$$\left\{x \to -\frac{8\,\text{ProductLog}\left[-\frac{1}{8}\,(-1)^{1/4}\,\text{Log}[4]\right]}{\text{Log}[4]}\right\},$$

$$\left\{x \to -\frac{8\,\text{ProductLog}\left[\frac{1}{8}\,(-1)^{1/4}\,\text{Log}[4]\right]}{\text{Log}[4]}\right\},$$

$$\left\{x \to -\frac{8\,\text{ProductLog}\left[-\frac{1}{8}\,(-1)^{3/4}\,\text{Log}[4]\right]}{\text{Log}[4]}\right\},$$

$$\left\{x \to -\frac{8\,\text{ProductLog}\left[\frac{1}{8}\,(-1)^{3/4}\,\text{Log}[4]\right]}{\text{Log}[4]}\right\},$$

$$\left.\left\{x \to -\frac{8\,\text{ProductLog}\left[-1, -\frac{\text{Log}[4]}{8}\right]}{\text{Log}[4]}\right\}\right\}$$

Problem 6

This problem demonstrates *Mathematica*'s **Limit** command.

- **(a)**

```
Limit[Tan[x] / x, x -> 0]
```

1

- **(b)**

```
Limit[1 / x, x → 0, Direction → -1]
```

∞

```
Limit[1 / x, x → 0, Direction → 1]
```

$-\infty$

- **(c)**

```
Limit[x Exp[-x^2], x -> Infinity]
```

0

```
Limit[x Exp[-x], x → -Infinity]
```

$-\infty$

- **(d)**

```
Limit[(Log[1 - x] + x) / x^2, x → 0]
```

$-\frac{1}{2}$

Problem 7

We find the derivatives of the given functions using *Mathematica*'s **D** command.

```
? D
```

D[f, x] gives the partial derivative $\partial f / \partial x$.
D[f, {x, n}] gives the multiple derivative $\partial^n f / \partial x^n$.
D[f, x, y, …] differentiates f successively with respect to x, y, ….
D[f, {{x_1, x_2, …}}] for a scalar f gives the vector derivative $(\partial f / \partial x_1, \partial f / \partial x_2, ...)$.
D[f, {*array*}] gives a tensor derivative. ≫

- **(a)**

In[5]:=
```
D[x^3 / (x^2 + 1), x]
```

Out[5]= $-\frac{2 x^4}{(1 + x^2)^2} + \frac{3 x^2}{1 + x^2}$

In[6]:=
```
Simplify[%]
```

Out[6]= $\frac{x^2 (3 + x^2)}{(1 + x^2)^2}$

- **(b)**

```
D[Sin[Sin[Sin[x]]], x]
```

```
Cos[x] Cos[Sin[x]] Cos[Sin[Sin[x]]]
```

- **(c)**

```
In[7]:= D[ArcTan[x], {x, 3}]
```

Out[7]= $$\frac{8\,x^2}{\left(1+x^2\right)^3}-\frac{2}{\left(1+x^2\right)^2}$$

```
In[8]:= Simplify[%]
```

Out[8]= $$\frac{-2+6\,x^2}{\left(1+x^2\right)^3}$$

- **(d)**

```
D[Sqrt[1 + x^2], x]
```

$$\frac{x}{\sqrt{1+x^2}}$$

- **(e)**

```
D[Exp[x * Log[x]], x]
```

$x^x\,(1+\mathrm{Log}[x])$

Problem 8

This problem demonstrates *Mathematica*'s **Integrate** command. We check some of the answers by differentiating. Here is a description of the command.

```
? Integrate
```

Integrate[f, x] gives the indefinite integral $\int f\, d x$.

Integrate[f, {x, x_{min}, x_{max}}] gives the definite integral $\int_{x_{min}}^{x_{max}} f\, d x$.

Integrate[f, {x, x_{min}, x_{max}}, {y, y_{min}, y_{max}}, ...]

gives the multiple integral $\int_{x_{min}}^{x_{max}} d x \int_{y_{min}}^{y_{max}} d y \ldots f$. ≫

- **(a)**

```
Integrate[Exp[-3 * x] * Sin[x], x]
```

$$-\frac{1}{10} e^{-3 x} (\text{Cos}[x] + 3\, \text{Sin}[x])$$

```
D[%, x]
```

$$-\frac{1}{10} e^{-3 x} (3\, \text{Cos}[x] - \text{Sin}[x]) + \frac{3}{10} e^{-3 x} (\text{Cos}[x] + 3\, \text{Sin}[x])$$

```
Simplify[%]
```

$$e^{-3 x}\, \text{Sin}[x]$$

- **(b)**

```
Integrate[(x + 1) * Log[x], x]
```

$$-x - \frac{x^2}{4} + x\, \text{Log}[x] + \frac{1}{2} x^2\, \text{Log}[x]$$

```
D[%, x]
```

$$\text{Log}[x] + x\, \text{Log}[x]$$

- **(c)**

```
Integrate[Sqrt[x / (1 - x)], {x, 0, 1}]
```

$$\frac{\pi}{2}$$

- **(d)**

```
Integrate[Exp[-x^2], {x, -Infinity, Infinity}]
```

$$\sqrt{\pi}$$

- **(e)**

```
Integrate[Sqrt[x^4 + 1], {x, 0, 1}]
```

$$\text{Hypergeometric2F1}\left[-\frac{1}{2}, \frac{1}{4}, \frac{5}{4}, -1\right]$$

```
N[%]
```

1.08943

Problem 9

```
Solve[x^5 - 3 * x^2 + x + 1 == 0, x]
```

$$\left\{\{x \to 1\}, \{x \to 1\}, \left\{x \to \frac{1}{3}\left(-2 - 5\left(\frac{2}{11 + 3\sqrt{69}}\right)^{1/3} + \left(\frac{1}{2}\left(11 + 3\sqrt{69}\right)\right)^{1/3}\right)\right\},\right.$$

$$\left\{x \to -\frac{2}{3} - \frac{1}{6}\left(1 + i\sqrt{3}\right)\left(\frac{1}{2}\left(11 + 3\sqrt{69}\right)\right)^{1/3} + \frac{5\left(1 - i\sqrt{3}\right)}{3 \times 2^{2/3}\left(11 + 3\sqrt{69}\right)^{1/3}}\right\},$$

$$\left.\left\{x \to -\frac{2}{3} - \frac{1}{6}\left(1 - i\sqrt{3}\right)\left(\frac{1}{2}\left(11 + 3\sqrt{69}\right)\right)^{1/3} + \frac{5\left(1 + i\sqrt{3}\right)}{3 \times 2^{2/3}\left(11 + 3\sqrt{69}\right)^{1/3}}\right\}\right\}$$

```
N[%]
```

$\{\{x \to 1.\}, \{x \to 1.\}, \{x \to -0.43016\},$
$\{x \to -0.78492 - 1.30714\, i\}, \{x \to -0.78492 + 1.30714\, i\}\}$

```
plot = Plot[x^5 - 3 * x^2 + x + 1, {x, -2, 2}]
```




```
Show[plot, PlotRange → {-5, 5}]
```



There is a root at approximately x = -0.43, a double root at x = 1, and the other two roots are complex.

Problem 10

```
f[t_] := t^6 - 4 * t^4 - 2 t^3 + 3 t^2 + 2 * t
```

- **(a)**

```
Plot[f[t], {t, -3 / 2, 5 / 2}]
```



- **(b)**

There is a local min near t = 1.6, a local max near t = 0.6, another local min near t = -0.4, and we cannot tell what is happening near t = -1.

```
Plot[f[t], {t, -1.2, -1 / 2}]
```



Might be an inflection point at t = -1.

- **(c)**

```
D[f[t], t]
```

$2 + 6\,t - 6\,t^2 - 16\,t^3 + 6\,t^5$

```
h[t_] := 2 + 6 t - 6 t^2 - 16 t^3 + 6 t^5
```

Alternatively, you can define **h[t_]:=Evaluate[D[f[t],t]]**,
but the command **Evaluate** is not discussed unitl Chapter 8 of *DEwMma*.

```
Plot[h[t], {t, -3 / 2, 5 / 2}]
```



Looks like there are four points where the derivative is zero as predicted by part (b).

```
FindRoot[h[t] == 0, {t, -1}]
```

```
{t → -1.}
```

```
FindRoot[h[t] == 0, {t, -0.4}]
```

```
{t → -0.314273}
```

```
FindRoot[h[t] == 0, {t, 0.6}]
```

```
{t → 0.629579}
```

```
FindRoot[h[t] == 0, {t, 1.6}]
```

```
{t → 1.68469}
```

- **(d)**

```
k[t_] := Evaluate[D[h[t], t]]
```

```
Plot[k[t], {t, -1.2, -0.8}]
```



This graph shows that t = -1 is indeed an inflection point because to the left it is convex up (the second derivative is positive) and to the right it is concave down (the second derivative is negative). Also t = -1 is an extreme point of the first derivative, corresponding to an inflection point of the function.

Problem 11

- **(a)**

```
Solve[{x^2 - y^2 == 1, 2 * x + y == 2}, {x, y}]
```

$$\left\{\{x \to 1, y \to 0\}, \left\{x \to \frac{5}{3}, y \to -\frac{4}{3}\right\}\right\}$$

- **(b)**

```
Plot1 := ContourPlot[2 * x + y == 2,
  {x, 0.5, 2}, {y, -2, 0.5}, ContourStyle -> Red]
```

```
Plot2 := ContourPlot[x^2 - y^2 == 1,
  {x, 0.5, 2}, {y, -2, 0.5}, ContourStyle -> Green]
```

```
Show[Plot1, Plot2]
```



Problem 12

- **(a)**

$$\sum_{n=1}^{\infty} 1 / n\hat{}2$$

$$\frac{\pi^2}{6}$$

- **(b)**

$$\sum_{n=0}^{\infty} x\hat{}n$$

$$\frac{1}{1 - x}$$

- **(c)**

```
f[x_] := Sum[x^n / n, {n, 1, ∞}]
```

$$f[x\_] := \sum_{n=1}^{\infty} x^n / n$$

```
f[x]
```

```
-Log[1 - x]
```

- **(d)**

```
D[f[x], x]
```

$$\frac{1}{1-x}$$

- **(e)**

$$\sum_{n=1}^{\infty} D[x^n / n, x]$$

$$-\frac{1}{-1+x}$$

This suggests that you can differentiate a power series, like a polynomial, term-by-term.

Problem 13

See Problem 3 for the syntax of the function **ContourPlot**:

Here’s a sample wherein it gives a collection of level curves:

```
ContourPlot[x^2 + y^2, {x, -5, 5}, {y, -5, 5}]
```



- **(a)**

```
ContourPlot[3 y + y^3 - x^3 == 5, {x, -10, 10}, {y, -10, 10}]
```



■ (b)

```
ContourPlot[{3 y + y^3 - x^3 == -2, 3 y + y^3 - x^3 == 0, 3 y + y^3 - x^3 == 2,
  3 y + y^3 - x^3 == 5, 3 y + y^3 - x^3 == 8}, {x, -5, 5}, {y, -5, 5}]
```



■ (c)

```
f[x_, y_] := y * Log[x] + x * Log[y]
```

```
f[1, 1]
```

```
0
```

```
ContourPlot[f[x, y] == f[1, 1], {x, -1, 5}, {y, -1, 5}]
```



Problem 14

```
u := Exp[-x] * Sin[x] + Exp[x] * Cos[x]
```

```
u
```

$e^x \mathrm{Cos}[x] + e^{-x} \mathrm{Sin}[x]$

- **(a)**

```
xpts = Table[j * Pi, {j, -1, 1, 1 / 4}]
```

$\left\{-\pi, -\frac{3\pi}{4}, -\frac{\pi}{2}, -\frac{\pi}{4}, 0, \frac{\pi}{4}, \frac{\pi}{2}, \frac{3\pi}{4}, \pi\right\}$

■ (b)

```
u /. x → xpts
```

$$\left\{-e^{-\pi},\ -\frac{e^{-3\pi/4}}{\sqrt{2}}-\frac{e^{3\pi/4}}{\sqrt{2}},\ -e^{\pi/2},\ \frac{e^{-\pi/4}}{\sqrt{2}}-\frac{e^{\pi/4}}{\sqrt{2}},\ 1,\ \frac{e^{-\pi/4}}{\sqrt{2}}+\frac{e^{\pi/4}}{\sqrt{2}},\ e^{-\pi/2},\ \frac{e^{-3\pi/4}}{\sqrt{2}}-\frac{e^{3\pi/4}}{\sqrt{2}},\ -e^{\pi}\right\}$$

■ (c)

```
f[x_] := Exp[-x] * Sin[x] + Exp[x] * Cos[x]
```

■ (d)

```
f[xpts]
```

$$\left\{-e^{-\pi},\ -\frac{e^{-3\pi/4}}{\sqrt{2}}-\frac{e^{3\pi/4}}{\sqrt{2}},\ -e^{\pi/2},\ \frac{e^{-\pi/4}}{\sqrt{2}}-\frac{e^{\pi/4}}{\sqrt{2}},\ 1,\ \frac{e^{-\pi/4}}{\sqrt{2}}+\frac{e^{\pi/4}}{\sqrt{2}},\ e^{-\pi/2},\ \frac{e^{-3\pi/4}}{\sqrt{2}}-\frac{e^{3\pi/4}}{\sqrt{2}},\ -e^{\pi}\right\}$$

```
N[%, 5]
```

{-0.043214, -7.5275, -4.8105, -1.2285,
1.0000, 1.8733, 0.20788, -7.3935, -23.141}

■ (e)

```
Table[{j * Pi, N[f[j * Pi], 5]}, {j, -1, 1, 1 / 4}] // TableForm
```

$-\pi$	-0.043214
$-\frac{3\pi}{4}$	-7.5275
$-\frac{\pi}{2}$	-4.8105
$-\frac{\pi}{4}$	-1.2285
0	1.0000
$\frac{\pi}{4}$	1.8733
$\frac{\pi}{2}$	0.20788
$\frac{3\pi}{4}$	-7.3935
π	-23.141